

Dynamic Verilog-A Model of a Magnetoresistive Spin Valve

A. Jander, L. Engelbrecht and P. Dhagat
 Oregon State University
 Corvallis, Oregon, USA
 jander@eecs.oregonstate.edu

ABSTRACT

A behavioral model for a magnetoresistive spin valve device implemented in Verilog-A is presented. The model realizes the complete static, single-domain magnetic hysteresis response as well as a damped transient response based on Landau-Lifshitz-Gilbert dynamics. The model will facilitate the design and analysis of complex spintronic or hybrid spintronic-electronic circuits in a simulation environment that is familiar to most circuit designers. The utility of the spin valve model is demonstrated in the analysis of a MRAM cell.

1. INTRODUCTION

Spintronic devices such as magnetoresistive spin valves and magnetic tunnel junctions are increasingly finding application in logic and random access memory products[1]. However, despite advances in these emerging device technologies, simulation tools that model their behavior in electronic circuits are still lacking. In this paper, a Verilog-A model for the most common spintronic device, the spin valve, is presented. Verilog-A is a behavioral description language that is incorporated into many commercial SPICE-based circuit simulation packages. Thus the development of such models will facilitate the design of complex spintronic as well as hybrid CMOS-spintronic circuits using a software environment that is familiar to most circuit designers.

The Verilog-A model of the spin valve is based on the physical principles governing magnetic materials and the magnetoresistive effect as summarized in Section 2. The Verilog-A model itself is described in Section 3. In Section 4, the operation of the model is illustrated in the analysis of a typical magnetic random access memory (MRAM) cell.

2. PHYSICAL PRINCIPLES OF SPIN VALVES

A spin valve consists of two magnetic layers, M_1 and M_2 , separated by a thin non-magnetic spacer layer as illustrated in Fig. 1. The magnetization of M_1 is free to rotate in response to an applied magnetic field, while M_2 is held fixed by an adjacent pinning layer. The resistance of the device depends on the relative orientation of the magnetic layers: lowest when the layers are magnetized in the same




Figure 1. Structure of a spin valve (inset) and its use in an MRAM cell. Labels rp, rn, blp, etc. correspond to port names of the MRAM module in Listing 1.

direction and highest when magnetized oppositely. The resistance is described by the function

$$R = R_{MIN} + \frac{1}{2}(R_{MAX} - R_{MIN})(1 - \cos(\theta_{M1} - \theta_{M2})) \quad (1)$$

where θ_{M1} and θ_{M2} are the angles of the magnetization in the free and pinned layers respectively. In contemporary devices, the magnetoresistance ratio, $\frac{R_{MAX} - R_{MIN}}{R_{MIN}}$ can be as high as 300%.

In a typical MRAM cell, the free layer magnetization is set by magnetic fields from two adjacent, orthogonal circuit traces, the *word* and *bit* lines shown in Fig. 1. Magnetic switching thresholds for M_1 are chosen such that the field from current in a single line is insufficient to switch the magnetization. Only the coincident application of current in both lines will produce enough field to switch the bit, allowing the selection of a single bit from a large array.

2.1 Magnetization Dynamics

The dynamic magnetic response of a small magnetic element such as the free layer is described by the Landau-Lifshitz-Gilbert (LLG) equation [2]

$$\frac{d\vec{M}}{dt} = \mu_0 \gamma \vec{M} \times \vec{H} + \frac{\alpha}{|\vec{M}|} \vec{M} \times \frac{d\vec{M}}{dt} \quad (2)$$

where \vec{M} , the magnetic moment and \vec{H} , the effective magnetic field are, in general, vectors in three dimensions,

γ is the gyromagnetic constant and α is the Gilbert damping parameter for the material.

For thin films, shape anisotropy forces the magnetization vector to lie principally in the plane and an effective in-plane magnetization dynamic can be found [3, 4]

$$\frac{d\vec{M}'}{dt} = -\mu_0\gamma\left(\alpha + \frac{1}{\alpha}\right)\vec{M}' \times (\vec{M}' \times \vec{H}') \quad (3)$$

for $\vec{M}' = (M_x, M_y)$ and $\vec{H}' = (H_x, H_y)$ restricted to the film plane. The magnetic elements are modeled as a single magnetic domain. Accordingly, the magnitude of \vec{M} remains constant ($|\vec{M}| \equiv M_s$) so that the vector \vec{M}' is fully described by its angle, θ_M , with respect to the x axis. The dynamic equation then takes the form

$$\frac{d\theta_M}{dt} = \mu_0\gamma\left(\alpha + \frac{1}{\alpha}\right)(H_y \cos\theta_M - H_x \sin\theta_M) \quad (4)$$

We now cast this equation equivalently as a sum of torques (more precisely, torque per unit volume) acting on the magnetic moment.

$$0 = \mu_0 M_s (H_y \cos\theta_M - H_x \sin\theta_M) - \frac{M_s}{\gamma(\alpha + \frac{1}{\alpha})} \frac{d\theta_M}{dt} \quad (5)$$

In this, the first two terms are the torque due to external fields and the final term is the damping torque.

For bistable operation, for example in an MRAM cell, a uniaxial anisotropy is required to hold the magnetization in one of two opposite directions. The anisotropy is modeled as an additional torque that pulls the magnetization towards the easy axis (positive *or* negative x-direction):

$$T_{anis} = -K_u \sin(2\theta_M) \quad (6)$$

where K_u is the strength of the uniaxial anisotropy. The balance of torques then becomes

$$0 = \mu_0 M_s (H_y \cos\theta_M - H_x \sin\theta_M) - K_u \sin(2\theta_M) - \frac{M_s}{\gamma(\alpha + \frac{1}{\alpha})} \frac{d\theta_M}{dt} \quad (7)$$

The first three terms describe the static behavior of a single domain particle and the final term captures the dynamic or transient response.

This physical model captures the full two-dimensional hysteretic switching characteristics and dynamic magnetization response of a single-domain, thin-film magnetic element with uniaxial anisotropy.




Figure 2. Block diagram of the modules composing the MRAM cell model.

3. VERILOG-A MODEL

The MRAM cell model is composed of three mixed-discipline Verilog-A modules as illustrated in Figure 2. The write line (WL) module determines the magnetic field in terms of the input current. The single domain (SD) module models the behavior of the magnetic element in response to the fields. Finally, the magnetoresistance (MR) module provides an output back in the electrical discipline based on the orientation of the free layer.

A new signal flow discipline, `sig_flow_H` with the access function `H()` is defined for magnetic field in units of A/m. The magnetization angle and torque are represented in the rotational discipline defined in the standard `disciplines.vams` file. Since the external connections to the MRAM super-module are all of the electrical discipline, the

Listing 1. Definitions and MRAM Cell top-level module

```

`include "constants.vams"
`include "disciplines.vams"

nature Magnetic_Field
  units = "A/m";
  access = H;
  abstol = 1p;
endnature

discipline sig_flow_H // signal-flow discipline
  potential Magnetic_Field;
enddiscipline

module MRAM_cell(wlp, wln, blp, bln, rp, rn);
  inout wlp, wln, blp, bln, // word and bit line connections
  inout rp, rn; // magnetoresistor connections
  electrical wlp, wln, blp, bln, rp, rn;
  rotational M1, M2; // free and pinned layer magnetization directions
  sig_flow_H hx, hy; // magnetic fields from write lines

  writeline bitline (.vp(blp), .vn(bln), .hwrite(hx));
  writeline wordline (.vp(wlp), .vn(wln), .hwrite(hy));
  single_domain freelayer (.hx(hx), .hy(hy), .M(M1));
  //Theta(M2) pinned to 0rads inside magnetoresistance analog process:
  magnetoresistance mr (.M1(M1), .M2(M2), .rp(rp), .rn(rn));
endmodule

```

module can easily be incorporated into a SPICE-based electrical circuit simulation.

As the primary purpose is to demonstrate the magnetic behavior of the spin valve, the write line and magnetoresistance modules are very simple.

3.1 Write Line (WL) Module

This simple write line model computes the magnetic field based on the current sheet approximation.

$$H = J_s / 2 \quad (8)$$

where J_s is the sheet current density in the line. This assumes that the magnetic element is close to a wide, thin conductor. The parameters W and R are the width and resistance of the write line respectively. More sophisticated models could include the effects of thicker lines or magnetically cladded lines and spacing loss due to the separation between the write line and the magnetic element. In most cases, however, the write field will still be simply proportional to the current.

Two write line modules are used in the MRAM cell, one for the word line, producing H_y and the other for the bit line producing H_x .

Listing 2. Write Line module

```

module writeline(vp,vn,hwrite);
inout vp, vn;
output hwrite;
electrical vp, vn;
sig_flow_H hwrite;

parameter real W = 1.0u; // write line width [m]
parameter real R = 0; // line resistance [ohms]

analog begin
    V(vp,vn) <+ R*I(vp,vn);
    H(hwrite) <+ I(vp,vn)/(2*W);
end
endmodule

```

3.2 Single Domain (SD) Magnetic Module

The single domain magnetic module describes the behavior of the magnetic element in response to the magnetic field components H_x and H_y . The input magnetic fields are in the newly defined sig_flow_H discipline. The magnetization angle and torque are represented in the rotational discipline using the access functions Theta(M) and Tau(M) respectively.

The analog block of the module implements equation (7). If the M output is not connected externally to an additional

Listing 3. Single Domain module

```

`define P_gamma 1.76e11 //Gyromagnetic constant [Hz/tesla]

module single_domain(hx, hy, M);
input hx, hy; // magnetic field vector components
inout M; // magnetization angle and torque
sig_flow_H hx, hy;
rotational M;

parameter real Ms = 8e5; // saturation magnetization [A/m]
parameter real Ku = 500; // uniaxial anisotropy [J/m^3]
parameter real alpha = 0.1; // LLG damping factor [unitless]

analog begin
    // torque due to external field:
    Tau(M) <+ -P_U0*Ms*(H(hx)*sin(Theta(M))-H(hy)*cos(Theta(M)));

    // anisotropy:
    Tau(M) <+ -Ku*sin(2*Theta(M));

    //damping torque:
    Tau(M) <+ -ddt(Theta(M))* Ms/((alpha+1/alpha)*P_gamma);
end
endmodule

```

source of torque, the Verilog-A compiler will look for solutions that result in zero net torque, since torque is the flow nature of the rotational discipline.

The parameters for the SD module are M_s , the saturation magnetization, K_u the uniaxial anisotropy and α , the damping parameter.

3.3 Magnetoresistance (MR) Module

The MR module models the magnetoresistance effect which determines the output resistance as a function of the magnetization angles of the two magnetic layers. In the case of the spin valve, one layer is held fixed by connecting port M2 externally to a constant angle source. A simple $\Delta R \propto \cos(\theta_{M1} - \theta_{M2})$ model is used in this example. A more complete model, such as the one described by

Listing 4. Magnetoresistance module

```

module magnetoresistance(M1,M2,rp,rn);
inout M1,M2,rp,rn;
rotational M1,M2;
electrical rp,rn;

parameter real R_max = 1000; // High resistance [ohm]
parameter real R_min = 500; // Low resistance [ohm]

analog begin
    V(rp,rn) <+ I(rp,rn)*(R_min+ 0.5*(R_max-R_min)*
        (1-cos(Theta(M1)-Theta(M2)))));
end
endmodule

```

Zhao *et al.*[5] would take into account the voltage dependence of the resistance.

The parameters for the MR module are the maximum and minimum resistance values of the device.

4. RESULTS

4.1 Hysteresis Loops

The behavior of the single domain module is demonstrated first. The model response for fields in the x direction, y direction and at 45° is shown in Figure 3. In each case, the field is swept positive and then negative. As is traditional, only the component of magnetization in the applied field direction is plotted. The resulting hysteretic curves are identical to those calculated by Stoner and Wohlfarth in their seminal 1948 work [6] on single-domain particles. The x-axis (easy axis) curve displays the expected square hysteresis loop while the y-axis (hard axis) response is hysteresis free. The 45° loop displays some hysteresis but with considerably lower switching field than the x-axis response. The spikes in the 45° degree loop are transients resulting from the time-stepped solution in Verilog.




Figure 3. Hysteresis loops generated with the Single Domain module.

4.2 Simulation of MRAM cell

A transient simulation for the full MRAM cell model is shown in Figure 4. A sequence of four current pulses is applied to the bit and word lines as shown in the upper plot. The lower plot shows the resulting change in resistance of the spin valve. The sequence of pulses is as follows: first, a large negative current pulse, exceeding the switching threshold is applied to the bit line to switch the free layer from the initial low resistance 0° to the high resistance state at 180°. The second pulse is a smaller positive current on the bit line which does not result in any




Figure 4. Simulation of MRAM cell writing showing resistance changes in response to current pulses on the bit and word lines

resistance change because it does not exceed the switching threshold. The third pulse is a current pulse on the word line, also below the switching threshold. This pulse results in a small, temporary rotation of the magnetization and a small change in resistance for the duration of the pulse. Finally, a coincident pulse on both the word and bit lines, resulting in a 45° field exceeding the switching threshold. The spin valve is switched backed to the original low resistance state by this pulse.

This sequence illustrates situations that would occur in the addressing of a MRAM array. The second and third pulses are those that would be seen by un-addressed bits on the same word or bit line as an addressed bit. The final pulse is that which would be experienced by the addressed bit. In addition to correctly modeling the switching, the resistance variations of the un-addressed bits are accurately represented.

The model also captures time delays associated with the damped response of the magnetization, particularly in situations where large angle rotations are called for. Particularly notable is the long delay from the beginning of the first field pulse to when the magnetization finally reverses itself. The field from the bit line in this case is exactly opposite of the original magnetization direction, resulting in an unstable equilibrium which persists for almost one nanosecond. In fact, a small 1μA current was required in the word line to get the spin valve to switch at all.

5. CONCLUSIONS

This simple model based on physical principles realizes the complex, hysteretic behavior of a single-domain magnetic particle.

The modular, physics-based approach to modeling of magnetic elements in Verilog-A is easy to extend to more complex devices. For example, a pseudo spin valve (in

which the second magnetic layer has a larger anisotropy but is not pinned as in a true spin valve) can be constructed using two SD models as illustrated in Figure 5. Additional modules could link the magnetization angles to simulate exchange coupling torques.

More complete write line models that include higher-order geometrical effects as well line impedance can easily be substituted for the WL module. Similarly, more sophisticated magnetoresistance models, which, for instance include bias voltage effects, can be incorporated into the MR module without affecting the single domain magnetic response.




Figure 5. Connection of the modules to simulate a pseudo spin valve device.

6. REFERENCES

- [1] C. Chappert, A. Fert, and F. N. V.D., "The emergence of spin electronics in data storage," *Nature Materials*, vol. 6, pp. 813-823, 2007.
- [2] L. Landau and E. Lifshitz, "Theory of the dispersion of magnetic permeability in ferromagnetic bodies," *Phys. Z. Sowjetunion*, vol. 8, p. 153, 1935.
- [3] J. G.-C. Carlos and E. Weinan, "Effective dynamics for ferromagnetic thin films," in *Journal of Applied Physics*, vol. 90, pp. 370-374 2001.
- [4] R. V. Kohn and V. V. Slustikov, "Effective dynamics for ferromagnetic thin films: a rigorous justification," *Proceedings of the Royal Society A*, vol. 461, pp. 143-154, 2005.
- [5] W. Zhao, E. Belhaire, Q. Mistral, C. Chappert, V. Javerliac, B. Dieny, and E. Nicolle, "Macro-model of Spin-Transfer Torque based Magnetic Tunnel Junction device for hybrid Magnetic-CMOS design," in *Behavioral Modeling and Simulation Workshop, Proceedings of the 2006 IEEE International*, 2006, pp. 40-43.
- [6] E. C. Stoner and E. P. Wohlfarth, "A Mechanism of Magnetic Hysteresis in Heterogeneous Alloys," *Philosophical Transactions of the Royal Society of London*, vol. A240, pp. 599-642, 1948.